//@version=5
indicator("Simplified ATR Trailing Stop", overlay=true)

// === Inputs ===
entryDateYear = input.int(2025, title="Entry Year", minval=1900, maxval=2100)
entryDateMonth = input.int(1, title="Entry Month", minval=1, maxval=12)
entryDateDay = input.int(1, title="Entry Day", minval=1, maxval=31)
entryPrice = input.float(0, title="Entry Price (Set to 0 for Close of the Day)", minval=0, maxval=5000)
atrLength = input.int(14, title="ATR Period", minval=1)
atrMultiplier = input.float(2.0, title="ATR Multiplier for Stop Loss", minval=0.1)
customValueForFirstDay = input.float(2.0, title="Custom Value for First Day Trailing Stop", minval=0.1, maxval=100)

// === ATR Calculation ===
atrValue = ta.atr(atrLength)

// === Entry Timestamp ===
entryDate = timestamp(entryDateYear, entryDateMonth, entryDateDay, 00, 00)

// === Entry Level ===
var float entryLevel = na
if (na(entryLevel) and time >= entryDate)
    entryLevel := (entryPrice > 0) ? entryPrice : close

// === Start Condition ===
startCalculating = (time >= entryDate)

// === Trailing Stop Logic ===
initialStop = entryLevel - atrMultiplier * atrValue
var float trailingStop = na
var bool firstDayCalculated = false
var bool stopPlotting = false  // NEW

if (startCalculating and not stopPlotting)
    if (na(trailingStop[1]))
        if (not firstDayCalculated)
            trailingStop := entryLevel - (atrValue * customValueForFirstDay)
            firstDayCalculated := true
        else
            trailingStop := initialStop
    else
        trailingStop := math.max(trailingStop[1], high - atrMultiplier * atrValue)

    // === NEW: Stop plotting if low < yesterday's trailing stop
    if (not na(trailingStop[1]) and low < trailingStop[1])
        stopPlotting := true

// === Plotting ===
plot(startCalculating and not stopPlotting ? trailingStop : na, color=color.red, title="Trailing Stop", linewidth=2)
